Modem connection fallback for eWON2005-Talk2M

1 Purpose

You have a site connected through Talk2M.

A common need is to have a backup connection (modem) when the main connection (through the corporate network or by eWON-ADSL) fails.

You can do that with an eWON2005 (or eWON4005) with a built-in modem (PSTN, ISDN or GPRS/EDGE) or with eWON-ADSL (2104 or 4104) with a GPRS modem.

For example:

- each site has a corporate network connection from the remote site to their HQ.
- normally, the eWON would make a VPN connection to Talk2M using the WAN port.
- if the corporate network line does not work, they need the eWON to make the VPN using the modem.
- when the corporate network starts to work again, the eWON drops the modem and uses the WAN again.

On topology where all remote sites use eWON-ADSL to go on Internet:

- each site has an ADSL connection from the remote site to their HQ.
- normally, the eWON would make a VPN connection to Talk2M using the ADSL port.
- if the ADSL line does not work, they need the eWON to make the VPN using the modem.
- when the ADSL line starts to work again, the eWON drops the modem and uses the ADSL again.

2 Principle

This kind of fallback can be done with a little BASIC script that simply switches from "Internet on WAN" (or "Internet on ADSL") to "Internet on Modem" when the corporate connection is down.

The following script monitors the VPN IP address (connected to Talk2M) and if this VPN-IP is 0.0.0.0 (due to the corporate network failure), the script switches the Internet connection mode to modem.

To return to the default connection mode, the script will simply wait an amount of time and switch back to WAN.
3 Tag creation

First you need to create a Tag WAN_ON that reflects the status of the WAN connection. This Boolean Tag has value 1 when the Ethernet connection is present.

Follow the path **Configuration > Tag setup > Create New Tag**

Enter WAN_ON in **Tag Name**, specify MEM as **Server Name** and set Boolean as **Type**:

![Tag creation interface](image)

Click **Add/Update**.

4 Script example

In the script below, we will test the VPN-IP every quarter (900 seconds), and when the fallback is activated, we will return to WAN after 2 hours (7200 seconds).

In the eWON communication configuration, you need to setup correctly the 2 communications ways and the Talk2M account.

**Rem**: the cyclic_section is empty and not shown below.

```
Rem --- eWON start section: Init Section
ewon_init_section:
Rem --- eWON user (start)
ONTIMER 1,"goto TestVPN"
TSET 1,900
WAN_ON@=1
ONTIMER 2,"goto ReturnToWAN"
TSET 2,0
Rem --- eWON user (end)
End
Rem --- eWON end section: Init Section
```
Rem --- eWON start section: Fallback
Rem --- eWON user (start)
TestVPN:
SETSYS INF,"load"
A$ = GETSYS INF,"VPNIP"
PRINT Time$;" ";A$
IF (A$="0.0.0.0") THEN
REM WAN is broken, activate Fallback
WAN_ON@=0
TSET 1,0
TSET 2,7200
SETSYS COM,"load"
SETSYS COM,"WanCnx",1
SETSYS COM,"save"
PRINT Time$;" Fallback activated"
LOGEVENT "Fallback activated"
ELSE
WAN_ON@=1
ENDIF
END
ReturnToWAN:
TSET 2,0
SETSYS COM,"load"
SETSYS COM,"WanCnx",2 : REM set value to 3 for eWON-ADSL
SETSYS COM,"save"
PRINT Time$;"Fallback deactivated"
LOGEVENT "Fallback deactivated"
TSET 1,900
END
Rem --- eWON user (end)
End
Rem --- eWON end section: Fallback